# $Id: PKGBUILD 182375 2013-04-09 18:35:29Z ioni $
# Maintainer : Ionut Biru <ibiru@archlinux.org>

pkgname=libva
pkgver=1.1.1
pkgrel=1
pkgdesc="Video Acceleration (VA) API for Linux"
arch=('i686' 'x86_64')
url="http://freedesktop.org/wiki/Software/vaapi"
license=('MIT')
depends=('libegl' 'libgl' 'libdrm' 'libxfixes')
makedepends=('mesa')
optdepends=('libva-vdpau-driver: vdpau back-end for nvidia'
            'libva-intel-driver: back-end for intel cards')
options=('!libtool')
source=(http://www.freedesktop.org/software/vaapi/releases/$pkgname/$pkgname-$pkgver.tar.bz2)
sha1sums=('755e488677a196c9ac18b8c0d11aa2a56e050306')

build() {
  cd "$pkgname-$pkgver"
  ./configure --prefix=/usr
  make
}

package() {
  cd "$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
  install -m644 -D COPYING "$pkgdir/usr/share/licenses/$pkgname/COPYING"
}

# vim:set ts=2 sw=2 et:
